

# Mathematical Operations and Elementary Functions

Julia provides a complete collection of basic arithmetic and bitwise operators across all of its numeric primitive types, as well as providing portable, efficient implementations of a comprehensive collection of standard mathematical functions.

## Arithmetic Operators

The following [arithmetic operators](#) are supported on all primitive numeric types:

Expression	Name	Description
$+x$	unary plus	the identity operation
$-x$	unary minus	maps values to their additive inverses
$x + y$	binary plus	performs addition
$x - y$	binary minus	performs subtraction
$x * y$	times	performs multiplication
$x / y$	divide	performs division
$x \div y$	integer divide	$x / y$ , truncated to an integer
$x \setminus y$	inverse divide	equivalent to $y / x$
$x ^ y$	power	raises $x$ to the $y$ th power
$x \% y$	remainder	equivalent to <code>rem(<math>x</math>, <math>y</math>)</code>

as well as the negation on [Bool](#) types:

Expression	Name	Description
$!x$	negation	changes true to false and vice versa

A numeric literal placed directly before an identifier or parentheses, e.g.  $2x$  or  $2(x+y)$ , is treated as a multiplication, except with higher precedence than other binary operations. See [Numeric Literal Coefficients](#) for details.

Julia's promotion system makes arithmetic operations on mixtures of argument types "just work" naturally and automatically. See [Conversion and Promotion](#) for details of the promotion system.

Here are some simple examples using arithmetic operators:

```
julia> 1 + 2 + 3
6

julia> 1 - 2
-1

julia> 3*2/12
0.5
```

(By convention, we tend to space operators more tightly if they get applied before other nearby operators. For instance, we would generally write  $-x + 2$  to reflect that first  $x$  gets negated, and then 2 is added to that result.)

When used in multiplication, `false` acts as a *strong zero*:

```
julia> NaN * false
0.0

julia> false * Inf
0.0
```

This is useful for preventing the propagation of NaN values in quantities that are known to be zero. See [Knuth \(1992\)](#) for motivation.

## Bitwise Operators

The following [bitwise operators](#) are supported on all primitive integer types:

Expression	Name
$\sim x$	bitwise not
$x \ \& \ y$	bitwise and

<code>x   y</code>	bitwise or
<code>x ⊕ y</code>	bitwise xor (exclusive or)
<code>x &gt;&gt;&gt; y</code>	<a href="#">logical shift</a> right
<code>x &gt;&gt; y</code>	<a href="#">arithmetic shift</a> right
<code>x &lt;&lt; y</code>	logical/arithmetic shift left

Here are some examples with bitwise operators:

```
julia> ~123
-124

julia> 123 & 234
106

julia> 123 | 234
251

julia> 123 ⊕ 234
145

julia> xor(123, 234)
145

julia> ~UInt32(123)
0xffffffff84

julia> ~UInt8(123)
0x84
```

## Updating operators

Every binary arithmetic and bitwise operator also has an updating version that assigns the result of the operation back into its left operand. The updating version of the binary operator is formed by placing a `=` immediately after the operator. For example, writing `x += 3` is equivalent to writing `x = x + 3`:

```
julia> x = 1
1
```

```
julia> x += 3
4

julia> x
4
```

The updating versions of all the binary arithmetic and bitwise operators are:

```
+= -= *= /= \= ÷= %= ^= &= |= ⊔= >>= >>= <<=
```

### Note

An updating operator rebinds the variable on the left-hand side. As a result, the type of the variable may change.

```
julia> x = 0x01; typeof(x)
UInt8

julia> x *= 2 # Same as x = x * 2
2

julia> typeof(x)
Int64
```

## Vectorized "dot" operators

For every binary operation like  $^$ , there is a corresponding "dot" operation  $.^$  that is *automatically* defined to perform  $^$  element-by-element on arrays. For example,  $[1, 2, 3]^3$  is not defined, since there is no standard mathematical meaning to "cubing" a (non-square) array, but  $[1, 2, 3].^3$  is defined as computing the elementwise (or "vectorized") result  $[1^3, 2^3, 3^3]$ . Similarly for unary operators like  $!$  or  $\sqrt{\phantom{x}}$ , there is a corresponding  $.\sqrt{\phantom{x}}$  that applies the operator elementwise.

```
julia> [1,2,3].^3
3-element Array{Int64,1}:
 1
 8
27
```

More specifically, a  $.^$  b is parsed as the "dot" call  $(^).(a, b)$ , which performs a [broadcast](#) operation:

it can combine arrays and scalars, arrays of the same size (performing the operation elementwise), and even arrays of different shapes (e.g. combining row and column vectors to produce a matrix). Moreover, like all vectorized "dot calls," these "dot operators" are *fusing*. For example, if you compute `2 .* A.^2 .+ sin.(A)` (or equivalently `@. 2A^2 + sin(A)`, using the `@.` macro) for an array `A`, it performs a *single* loop over `A`, computing  $2a^2 + \sin(a)$  for each element of `A`. In particular, nested dot calls like `f.(g.(x))` are fused, and "adjacent" binary operators like `x .+ 3 .* x.^2` are equivalent to nested dot calls `(+).(x, (*).(3, (^).(x, 2)))`.

Furthermore, "dotted" updating operators like `a .+= b` (or `@. a += b`) are parsed as `a .= a .+ b`, where `.=` is a fused *in-place* assignment operation (see the [dot syntax documentation](#)).

Note the dot syntax is also applicable to user-defined operators. For example, if you define `⊗(A, B) = kron(A, B)` to give a convenient infix syntax `A ⊗ B` for Kronecker products ([kron](#)), then `[A, B] .⊗ [C, D]` will compute `[A⊗C, B⊗D]` with no additional coding.

Combining dot operators with numeric literals can be ambiguous. For example, it is not clear whether `1.+x` means `1. + x` or `1 .+ x`. Therefore this syntax is disallowed, and spaces must be used around the operator in such cases.

## Numeric Comparisons

Standard comparison operations are defined for all the primitive numeric types:

Operator	Name
<code>==</code>	equality
<code>!=, ≠</code>	inequality
<code>&lt;</code>	less than
<code>&lt;=, ≤</code>	less than or equal to
<code>&gt;</code>	greater than
<code>&gt;=, ≥</code>	greater than or equal to

Here are some simple examples:

```
julia> 1 == 1
true
```

```
julia> 1 == 2
false

julia> 1 != 2
true

julia> 1 == 1.0
true

julia> 1 < 2
true

julia> 1.0 > 3
false

julia> 1 >= 1.0
true

julia> -1 <= 1
true

julia> -1 <= -1
true

julia> -1 <= -2
false

julia> 3 < -0.5
false
```

Integers are compared in the standard manner – by comparison of bits. Floating-point numbers are compared according to the [IEEE 754 standard](#):

- Finite numbers are ordered in the usual manner.
- Positive zero is equal but not greater than negative zero.
- Inf is equal to itself and greater than everything else except NaN.
- -Inf is equal to itself and less than everything else except NaN.
- NaN is not equal to, not less than, and not greater than anything, including itself.

The last point is potentially surprising and thus worth noting:

```
julia> NaN == NaN
false
```

```
julia> NaN != NaN
true

julia> NaN < NaN
false

julia> NaN > NaN
false
```

and can cause headaches when working with [arrays](#):

```
julia> [1 NaN] == [1 NaN]
false
```

Julia provides additional functions to test numbers for special values, which can be useful in situations like hash key comparisons:

Function	Tests if
<code>isequal(x, y)</code>	x and y are identical
<code>isfinite(x)</code>	x is a finite number
<code>isinf(x)</code>	x is infinite
<code>isnan(x)</code>	x is not a number

`isequal` considers NaNs equal to each other:

```
julia> isequal(NaN, NaN)
true

julia> isequal([1 NaN], [1 NaN])
true

julia> isequal(NaN, NaN32)
true
```

`isequal` can also be used to distinguish signed zeros:

```
julia> -0.0 == 0.0
true
```

```
julia> isequal(-0.0, 0.0)
false
```

Mixed-type comparisons between signed integers, unsigned integers, and floats can be tricky. A great deal of care has been taken to ensure that Julia does them correctly.

For other types, `isequal` defaults to calling `==`, so if you want to define equality for your own types then you only need to add a `==` method. If you define your own equality function, you should probably define a corresponding `hash` method to ensure that `isequal(x, y)` implies `hash(x) == hash(y)`.

## Chaining comparisons

Unlike most languages, with the [notable exception of Python](#), comparisons can be arbitrarily chained:

```
julia> 1 < 2 <= 2 < 3 == 3 > 2 >= 1 == 1 < 3 != 5
true
```

Chaining comparisons is often quite convenient in numerical code. Chained comparisons use the `&&` operator for scalar comparisons, and the `&` operator for elementwise comparisons, which allows them to work on arrays. For example, `0 .< A .< 1` gives a boolean array whose entries are true where the corresponding elements of `A` are between 0 and 1.

Note the evaluation behavior of chained comparisons:

```
julia> v(x) = (println(x); x)
v (generic function with 1 method)

julia> v(1) < v(2) <= v(3)
2
1
3
true

julia> v(1) > v(2) <= v(3)
2
1
false
```

The middle expression is only evaluated once, rather than twice as it would be if the expression were written as `v(1) < v(2) && v(2) <= v(3)`. However, the order of evaluations in a chained comparison is undefined. It is strongly recommended not to use expressions with side effects (such as printing) in



chained comparisons. If side effects are required, the short-circuit `&&` operator should be used explicitly (see [Short-Circuit Evaluation](#)).

## Elementary Functions

Julia provides a comprehensive collection of mathematical functions and operators. These mathematical operations are defined over as broad a class of numerical values as permit sensible definitions, including integers, floating-point numbers, rationals, and complex numbers, wherever such definitions make sense.

Moreover, these functions (like any Julia function) can be applied in "vectorized" fashion to arrays and other collections with the [dot syntax](#) `f.(A)`, e.g. `sin.(A)` will compute the sine of each element of an array `A`.

## Operator Precedence and Associativity

Julia applies the following order and associativity of operations, from highest precedence to lowest:

Category	Operators	Associativity
Syntax	<code>.</code> followed by <code>::</code>	Left
Exponentiation	<code>^</code>	Right
Unary	<code>+</code> <code>-</code> <code>√</code>	Right <sup>[1]</sup>
Bitshifts	<code>&lt;&lt;</code> <code>&gt;&gt;</code> <code>&gt;&gt;&gt;</code>	Left
Fractions	<code>//</code>	Left
Multiplication	<code>*</code> <code>/</code> <code>%</code> <code>&amp;</code> <code>\</code> <code>÷</code>	Left <sup>[2]</sup>
Addition	<code>+</code> <code>-</code> <code> </code> <code>⊔</code>	Left <sup>[2]</sup>
Syntax	<code>:</code> <code>..</code>	Left
Syntax	<code> &gt;</code>	Left
Syntax	<code>&lt; </code>	Right
Comparisons	<code>&gt;</code> <code>&lt;</code> <code>&gt;=</code> <code>&lt;=</code> <code>==</code> <code>===</code> <code>!=</code> <code>!==</code> <code>&lt;:</code>	Non-associative

Control flow	&& followed by    followed by ?	Right
Pair	=>	Right
Assignments	= += -= *= /= // = \= ^= ÷= %=  = &= \_ = <<= >>= >>>=	Right

For a complete list of every Julia operator's precedence, see the top of this file: <src/julia-parser.scm>

[Numeric literal coefficients](#), e.g.  $2x$ , are treated as multiplications with higher precedence than any other binary operation, and also have higher precedence than  $^$ .

You can also find the numerical precedence for any given operator via the built-in function `Base.operator_precedence`, where higher numbers take precedence:

```
julia> Base.operator_precedence(:+), Base.operator_precedence(:*), Base.operator_precedence(:sin)
(11, 12, 17)

julia> Base.operator_precedence(:+), Base.operator_precedence(:sin), Base.operator_precedence(:left)
(11, 17, 0)
```

A symbol representing the operator associativity can also be found by calling the built-in function `Base.operator_associativity`:

```
julia> Base.operator_associativity(:-), Base.operator_associativity(:+), Base.operator_associativity(:left)
(:right, :none, :right)

julia> Base.operator_associativity(:*), Base.operator_associativity(:sin), Base.operator_associativity(:left)
(:right, 0, :right)
```

Note that symbols such as `:sin` return precedence 0. This value represents invalid operators and not operators of lowest precedence. Similarly, such operators are assigned associativity `:none`.

## Numerical Conversions

Julia supports three forms of numerical conversion, which differ in their handling of inexact conversions.

- The notation `T(x)` or `convert{T}(x)` converts `x` to a value of type `T`.
  - If `T` is a floating-point type, the result is the nearest representable value, which could be positive or negative infinity.

- If  $T$  is an integer type, an `InexactError` is raised if  $x$  is not representable by  $T$ .
- $x \% T$  converts an integer  $x$  to a value of integer type  $T$  congruent to  $x$  modulo  $2^n$ , where  $n$  is the number of bits in  $T$ . In other words, the binary representation is truncated to fit.
- The [Rounding functions](#) take a type  $T$  as an optional argument. For example, `round{Int}(x)` is a shorthand for `Int(round(x))`.

The following examples show the different forms.

```
julia> Int8(127)
127

julia> Int8(128)
ERROR: InexactError: trunc{Int8, 128}
Stacktrace:
[...]

julia> Int8(127.0)
127

julia> Int8(3.14)
ERROR: InexactError: Int8{3.14}
Stacktrace:
[...]

julia> Int8(128.0)
ERROR: InexactError: Int8{128.0}
Stacktrace:
[...]

julia> 127 % Int8
127

julia> 128 % Int8
-128

julia> round{Int8}(127.4)
127

julia> round{Int8}(127.6)
ERROR: InexactError: trunc{Int8, 128.0}
Stacktrace:
[...]
```

See [Conversion and Promotion](#) for how to define your own conversions and promotions.

## Rounding functions

Function	Description	Return type
<code>round(x)</code>	round $x$ to the nearest integer	<code>typeof(x)</code>
<code>round(T, x)</code>	round $x$ to the nearest integer	<code>T</code>
<code>floor(x)</code>	round $x$ towards $-\text{Inf}$	<code>typeof(x)</code>
<code>floor(T, x)</code>	round $x$ towards $-\text{Inf}$	<code>T</code>
<code>ceil(x)</code>	round $x$ towards $+\text{Inf}$	<code>typeof(x)</code>
<code>ceil(T, x)</code>	round $x$ towards $+\text{Inf}$	<code>T</code>
<code>trunc(x)</code>	round $x$ towards zero	<code>typeof(x)</code>
<code>trunc(T, x)</code>	round $x$ towards zero	<code>T</code>

## Division functions

Function	Description
<code>div(x, y)</code> , $x \div y$	truncated division; quotient rounded towards zero
<code>fld(x, y)</code>	floored division; quotient rounded towards $-\text{Inf}$
<code>cld(x, y)</code>	ceiling division; quotient rounded towards $+\text{Inf}$
<code>rem(x, y)</code>	remainder; satisfies $x == \text{div}(x, y) * y + \text{rem}(x, y)$ ; sign matches $x$
<code>mod(x, y)</code>	modulus; satisfies $x == \text{fld}(x, y) * y + \text{mod}(x, y)$ ; sign matches $y$
<code>mod1(x, y)</code>	mod with offset 1; returns $r \in (0, y]$ for $y > 0$ or $r \in [y, 0)$ for $y < 0$ , where $\text{mod}(r, y) == \text{mod}(x, y)$
<code>mod2pi(x)</code>	modulus with respect to $2\pi$ ; $0 \leq \text{mod2pi}(x) < 2\pi$
<code>divrem(x, y)</code>	returns $(\text{div}(x, y), \text{rem}(x, y))$
<code>fldmod(x, y)</code>	returns $(\text{fld}(x, y), \text{mod}(x, y))$

`gcd(x, y...)`    greatest positive common divisor of  $x, y, \dots$

`lcm(x, y...)`    least positive common multiple of  $x, y, \dots$

## Sign and absolute value functions

Function	Description
<code>abs(x)</code>	a positive value with the magnitude of $x$
<code>abs2(x)</code>	the squared magnitude of $x$
<code>sign(x)</code>	indicates the sign of $x$ , returning -1, 0, or +1
<code>signbit(x)</code>	indicates whether the sign bit is on (true) or off (false)
<code>copysign(x, y)</code>	a value with the magnitude of $x$ and the sign of $y$
<code>flipsign(x, y)</code>	a value with the magnitude of $x$ and the sign of $x*y$

## Powers, logs and roots

Function	Description
<code>sqrt(x)</code> , $\sqrt{x}$	square root of $x$
<code>cbrt(x)</code> , $\sqrt[3]{x}$	cube root of $x$
<code>hypot(x, y)</code>	hypotenuse of right-angled triangle with other sides of length $x$ and $y$
<code>exp(x)</code>	natural exponential function at $x$
<code>expm1(x)</code>	accurate $\exp(x) - 1$ for $x$ near zero
<code>ldexp(x, n)</code>	$x * 2^n$ computed efficiently for integer values of $n$
<code>log(x)</code>	natural logarithm of $x$
<code>log(b, x)</code>	base $b$ logarithm of $x$
<code>log2(x)</code>	base 2 logarithm of $x$

<code>log10(x)</code>	base 10 logarithm of $x$
<code>log1p(x)</code>	accurate $\log(1+x)$ for $x$ near zero
<code>exponent(x)</code>	binary exponent of $x$
<code>significand(x)</code>	binary significand (a.k.a. mantissa) of a floating-point number $x$

For an overview of why functions like `hypot`, `expm1`, and `log1p` are necessary and useful, see John D. Cook's excellent pair of blog posts on the subject: `expm1`, `log1p`, `erfc`, and `hypot`.

## Trigonometric and hyperbolic functions

All the standard trigonometric and hyperbolic functions are also defined:

```
sin    cos    tan    cot    sec    csc
sinh   cosh   tanh   coth   sech   csch
asin   acos   atan   acot   asec   acsc
asinh  acosh  atanh  acoth  asech  acsch
sinc   cosc
```

These are all single-argument functions, with `atan` also accepting two arguments corresponding to a traditional `atan2` function.

Additionally, `sinpi(x)` and `cospi(x)` are provided for more accurate computations of `sin(pi*x)` and `cos(pi*x)` respectively.

In order to compute trigonometric functions with degrees instead of radians, suffix the function with `d`. For example, `sind(x)` computes the sine of  $x$  where  $x$  is specified in degrees. The complete list of trigonometric functions with degree variants is:

```
sind   cosd   tand   cotd   secd   cscd
asind  acosd  atand  acotd  asecd  acscd
```

## Special functions

Many other special mathematical functions are provided by the package `SpecialFunctions.jl`.

- 1 The unary operators `+` and `-` require explicit parentheses around their argument to disambiguate them from the operator `++`, etc. Other compositions of unary operators are parsed with right-associativity, e. g.,  $\sqrt{\sqrt{-a}}$  as  $\sqrt{(\sqrt{-a})}$ .

- 2 The operators `+`, `++` and `*` are non-associative. `a + b + c` is parsed as `+(a, b, c)` not `+(+(a, b), c)`. However, the fallback methods for `+(a, b, c, d...)` and `*(a, b, c, d...)` both default to left-associative evaluation.

---

[« Integers and Floating-Point Numbers](#)[Complex and Rational Numbers »](#)

Powered by [Documenter.jl](#) and the [Julia Programming Language](#).